

Caso práctico

La tarea consiste en escribir una transformación **XSLT** que a partir de un fichero **XML** con datos de artistas genere una página web con una tabla con los datos de los artistas.

El elemento raíz del fichero es **artistas**. Dentro de este elemento hay uno o más elementos **artista**, como se puede ver en el ejemplo. Todos los elementos y atributos son obligatorios, salvo el elemento **fallecimiento**.

Al aplicar la transformación, el resultado debe ser una página web válida. Es decir, tiene que tener todos los elementos básicos de una página web: título, codificación de caracteres, elementos, html, body y head.

Dentro del cuerpo de la página habrá una tabla, obtenida a partir de los datos del fichero `artistas.xml`.

- La tabla tendrá una fila por cada artista mostrado, y una primera fila como cabecera, con los nombres: Código, Nombre, Año de nacimiento, Año de fallecimiento, País y Página web.
- Como se puede ver en el fichero, el año de fallecimiento no está disponible para todos los artistas. En ese caso, en la columna correspondiente se mostrará el texto “Desconocido”.
- En la última columna, el valor se obtiene del elemento `fichaCompleta`. Debe mostrarse un vínculo a la dirección adecuada, con el texto “Saber más”.
- Sólo se mostrarán los artistas nacidos después de 1500.
- Los artistas se mostrarán ordenados por año de nacimiento creciente. Es decir, en la primera fila, el que tenga un año de nacimiento anterior.

Lo primero que hice fue crear el archivo .xslt y enlazarlo en el xml añadiendo la siguiente línea al archivo xml:

```
<?xml-stylesheet type="html/xsl" href="Tarea5.xslt"?>
```

Ahora que el xml ya tiene vinculado el archivo de estilo xslt, creamos el xslt con el nombre que nos solicita el enunciado y pasamos a editarlo.

Lo primero, crear la declaración xml y la declaración xslt junto al espacio de nombres:

```
<?xml version="1.0" encoding="UTF-8"?>  
<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">
```

Añadimos también la configuración de la transformación la cual será un archivo html y con codificación UTF-8:

```
<xsl:output method="html" encoding="UTF-8"/>
```

Y ahora empezaremos a crear verdaderamente la transformación. Añadimos la siguiente línea para definir que transformación realizaremos con cada elemento “Artistas” el cual es elemento raíz.

```
<xsl:template match="/artistas">
```

Y a continuación escribimos un código que contenga las secciones principales de un documento html, tal y como se nos pide en el ejercicio. Añadimos las secciones head y body. Dentro de body, creamos una estructura de tabla con el cabecero de la tabla que se nos pide en el enunciado:

```
<h1>Lista de Artistas</h1>  
  <table border="0">  
    <tr>  
      <th>Código</th>  
      <th>Nombre</th>  
      <th>Año de nacimiento</th>  
      <th>Año de fallecimiento</th>  
      <th>País</th>  
      <th>Página web</th>  
    </tr>
```

A continuación se le dice que se aplique, para cada elemento “artista” la transformación que definiremos a continuación y aquí es donde especificamos que lo aplique solo a aquellos donde el campo “nacimiento” sea mayor que 1500 y también indicamos que lo ordene en sentido ascendente según el valor del campo “nacimiento”.

Esto se hace mediante la siguiente línea:

```
<xsl:apply-templates select="artista[nacimiento > 1500]">
  <xsl:sort select="nacimiento"/>
</xsl:apply-templates>
```

De esta manera, primero se creará el elemento fijo que es la cabecera de la tabla y a continuación recorrerá uno por uno los elementos artista aplicando las transformaciones que a continuación le indicamos, “para cada elemento artista...”:

```
<xsl:template match="artista">
```

Crea una fila con las siguientes columnas y en cada una de ellas muestra los siguientes datos:

```
<tr>
  <td><xsl:value-of select="@cod"/></td>
  <td><xsl:value-of select="nombreCompleto"/></td>
  <td><xsl:value-of select="nacimiento"/></td>
  <td>
    <xsl:choose>
      <xsl:when test="fallecimiento">
        <xsl:value-of select="fallecimiento"/>
      </xsl:when>
      <xsl:otherwise>Desconocido</xsl:otherwise>
    </xsl:choose>
  </td>
  <td><xsl:value-of select="pais"/></td>
  <td>
    <a href="{fichaCompleta}" target="_blank">Saber más</a>
  </td>
</tr>
```

En el caso de la fecha de fallecimiento, se nos dice que no es obligatoria por lo que puede aparecer o no, para este caso hemos usado “choose” para que en caso de que exista “fallecimiento” nos muestre el dato y de lo contrario, se nos muestre la palabra “Desconocido”


```
<xsl:choose>
  <xsl:when test="fallecimiento">
    <xsl:value-of select="fallecimiento"/>
  </xsl:when>
  <xsl:otherwise>Desconocido</xsl:otherwise>
</xsl:choose>
```

Por último, ya solo nos queda añadir el enlace a cada una de las webs “fichaCompleta” y que se muestre una etiqueta con la frase “Saber mas”:

```
<a href="{fichaCompleta}" target="_blank">Saber más</a>
```

Aunque también se valoró la posibilidad de hacerlo todo dentro desarrollado dentro del elemento “body”, decidí sacar fuera la transformación de cada elemento “artista” para mejor claridad del código, de esta manera vemos primero lo que generará la transformación y luego lo que debe hacer para cada fila de la tabla.

Este es el resultado que se obtiene:



Lista de Artistas

Código	Nombre	Año de nacimiento	Año de fallecimiento	País	Página web
a102	Michelangelo Caravaggio	1571	Desconocido	Italia	Saber más
a101	Diego Velázquez	1599	1660	España	Saber más
a104	Francisco de Goya	1746	1828	España	Saber más